
Calculate a Radiation from Two Offsets

Programmer: Dr. Bill Hazelton

Date: May, 2007.

Line Instruction Display User Instructions
Q0001 LBL Q
Q0002 SF 10  FLAGS  SF  .0
Q0003 ENTER RIGHT AZ  EQN  RCL E  RCL N etc. ENTER to end
Q0004 PSE
Q0005 INPUT  R
Q0006 RCL  R
Q0007 →HR
Q0008 STO  A
Q0009 ENTER LEFT AZ  EQN  RCL E  RCL N etc. ENTER to end
Q0010 PSE
Q0011 INPUT  L
Q0012 RCL  A
Q0013 RCL  L
Q0014 →HR
Q0015 –
Q0016 STO  E
Q0017 ENTER RIGHT OS  EQN  RCL E  RCL N etc. ENTER to end
Q0018 PSE
Q0019 INPUT  R
Q0020 ENTER LEFT OS  EQN  RCL E  RCL N etc. ENTER to end
Q0021 PSE
Q0022 INPUT  L
Q0023 RCL  R
Q0024 RCL  E
Q0025 SIN
Q0026 ×
Q0027 RCL  E
Q0028 COS
Q0029 RCL×  R
Q0030 RCL+  L
Q0031 ÷
Q0032 ATAN
Q0033 STO  C
Q0034 RCL  A
Q0035 x < > y
Q0036 –
Q0037 →HMS
Q0038 STO  B
Q0039 RADIATION AZ  EQN  RCL R  RCL A etc. ENTER to end

Q0040 PSE
Q0041 VIEW  B
Q0042 RCL  R
Q0043 RCL  C
Q0044 SIN
Q0045 ÷
Q0046 STO  D
Q0047 RAD LENGTH  EQN  RCL R  RCL A etc. ENTER to end
Q0048 PSE
Q0049 VIEW  D
Q0050 CF  10  FLAGS  CF  .0
Q0051 RTN

Notes

(1) This program allows a radiation (azimuth and distance) to an object to be
computed from two offsets measured from two lines of known azimuth to the
object.

(2) The two lines from which the offsets are measured do not have to be orthogonal.
The closer they are to orthogonal, the better, but the program will work with any
realistic set of measurements. It will also work across the 0° line, and with
traverse lines in different quadrants.

(3) The purpose of the program is to facilitate conversion of older survey data, in
which corners were often located by pairs of offsets, to a form in which a
radiation is employed. This will simplify calculations based on traverses and
radiations, i.e., vectors.

(4) Azimuths are entered and displayed in HP notation, i.e., DDD.MMSS, at all
times.

(5) Feet or meters (or any other linear units) can be used, provided their use is
consistent.

(6) The original code for this program was developed by Philip R. Price at the State
Rivers and Water Supply Commission (SR&WSC) Survey Branch, Victoria,
Australia, in November, 1976, for the HP-25 calculator. This program is an
update and adaptation for the HP-33S, but is based on Phil Price’s original
solution and HP-25 implementation.

Theory

If an object, commonly a corner, is located by offsets, these can be converted to the equivalent
radiation using the program. In the above example, the corner was located by two offsets, being
5.523 on the right and 25.408 on the left. The offsets are measured at right angles to the
respective traverse line, and are the distance between the object and the traverse line. The right
traverse line has an azimuth of 91° 56', while the left traverse line has an azimuth of 340° 36'. See
the figure on the next page.

Generalizing the diagram and giving it symbols, the general situation becomes as follows.

The angle θ is the difference between the two known azimuths, Azleft and Azright.. The offset from
the right traverse line to the object is dright, while the offset from the left traverse line to the object
is dleft. Because the offsets are measured at right angles to the traverse lines, their azimuths are
also known.

The angle α is the angle between the right traverse line and the radiation, and is determined using
the following formula:

� 

α = arctan
dright sin θ

dleft + dright cos θ
⎛ 

⎝ ⎜ 
⎞ 

⎠ ⎟ 

Subtracting α from Azright gives the azimuth of the radiation, Azrad. The length of the radiation,
drad, is derived using:

� 

drad =
dright
sinα

Precision of Results
Note that the precision of the derived azimuth and distance is limited by the precision with which
the offsets were measured, as well as the geometry of the offsets and radiation. The program
provides the one solution given the data, but without redundant measurements the precision of the
solution is unknown. It would be unwise to use azimuths for the radiation that are more precise
that about one minute of arc. Experiment with changing the values of the offsets by small
amounts that are consistent with their probable precision, and see what happens to the azimuth
and distance values for the radiation.

Sample Computations

1. Using the example given, where the right azimuth is 91° 56', the left azimuth is 340° 36',
the right offset is 5.523 and the left offset is 25.408, the radiation has an azimuth of 79°
32' and a distance of 25.72.

2. Using a right azimuth of 160° 36' and a left azimuth of 91° 56', a right offset of 13.272
and a left offset of 15.693, the radiation has an azimuth of 129° 32' and a distance of
25.72.

3. Using a right azimuth of 268° 35', a left azimuth of 195° 12', a right offset of 11.782 and
a left offset of 9.467, the radiation has an azimuth of 227° 15' and a distance of 17.84.

4. Using a right azimuth of 340° 36', a left azimuth of 271° 56', a right offset of 11.57 and a
left offset of 8.78, the radiation has an azimuth of 300° 55' and a distance of 18.12.

Running the Program
With everything to hand, press XEQ  Q.

The calculator briefly displays ENTER RIGHT AZ, then prompts R?

Key in the value of the azimuth of the right traverse line, in HP notation (DDD.MMSS).
Press R/S.

The calculator briefly displays ENTER LEFT AZ, then prompts L?

Key in the value of the azimuth of the left traverse line, in HP notation (DDD.MMSS).
Press R/S.

The calculator briefly displays ENTER RIGHT OS, then prompts R?

Key in the value of the right offset. Press R/S.

The calculator briefly displays ENTER LEFT OS, then prompts L?

Key in the value of the left offset. Press R/S.

The calculator briefly displays RADIATION AZ, then shows B= and the azimuth in HP notation
(DDD.MMSS). Press R/S.

The calculator briefly displays RAD LENGTH, then shows D= and the length of the radiation.

Press R/S to complete the program and clear the flag set at the start of the program.
To compute additional radiations from offset, press XEQ  Q and start the program again.

Storage Registers Used

A Azimuth of right traverse line (in decimal degrees)
B Azimuth of the radiation (in HP notation, DDD.MMSS)
C α, the angle between the right traverse line and the radiation (in decimal degrees)
D Length of the radiation
E θ, the angle between the traverse lines (in decimal degrees)
L Left traverse azimuth (in HP notation) or left offset (temporary storage)
R Right traverse azimuth (in HP notation) or right offset (temporary storage)

Labels Used

Label Q Length  =   229 Checksum  =  F54A

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computations to check proper operation after entry.


